Espacios vectoriales. Ortogonalizacioón de Gram−Schmidt

Práctica de Algebra Lineal, E.U.A.T., Curso 2005−2006

En esta práctica se explica cómo resolver con *Mathematica* algunos problemas comunes que aparecen al estudiar espacios vectoriales.

Dependencia e independencia lineal de vectores

■ Dado un conjunto de vectores, ¿cómo ver si son linealmente independientes o no?

■ Un ejemplo

Por ejemplo, supongamos que tenemos los vectores (1,1,0), (0,1,0). ¿Son linealmente independientes? (Es fácil de ver a ojo, pero hagámoslo con el ordenador como ejemplo.) Si escribimos la matriz formada por esos vectores (puestos en filas o en columnas, no importa), entonces el número de vectores independientes es el rango de la matriz. Sabemos calcular el rango de la matriz de otras prácticas:

```
In[1]:= M1 = {{1, 1, 0}, {0, 1, 0}}

Out[1]= {{1, 1, 0}, {0, 1, 0}}

In[2]:= MatrixForm[RowReduce[M1]]

Out[2]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \end{pmatrix}$$

El rango de esta matriz es 2, luego los dos vectores del principio son linealmente independientes (si el rango fuera menor que 2, eso nos diría que no lo son).

■ Otro ejemplo

Los vectores (1,1,0), (0,1,0), (1,2,0), ¿son linealmente independientes?
Para contestar esto hacemos lo mismo que antes:

```
In[3]:= M2 = {{1, 1, 0}, {0, 1, 0}, {1, 2, 0}}

Out[3]= {{1, 1, 0}, {0, 1, 0}, {1, 2, 0}}

In[4]:= MatrixForm[RowReduce[M2]]

Out[4]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

El rango de esta matriz es 2, no 3, luego los tres vectores del principio no son linealmente independientes. Además, sabemos del ejemplo anterior que los dos primeros vectores sí son independientes, así que debe ocurrir que (1,2,0) sea una combinación lineal de los otros dos. De hecho es fácil ver que el tercero es la suma de los dos primeros. (Observa que en este ejemplo cualquier par de vectores es linealmente independiente, pero siempre podemos escribir el otro como combinación lineal de los dos que se escojan.)

Nota: la orden **RowReduce** puede cambiar el orden de las filas de la matriz de partida y por eso no sirve directamente para decidir cuáles de los vectores iniciales son independientes.

■ Cambio de base

Si tenemos la coordenadas de un vector en una base **B1**, ¿cómo podemos saber las coordenadas del mismo vector en otra base B2? Una forma de hacerlo es usar la matriz de cambio de base de **B1** a **B2**, que es la matriz que tiene por columnas las coordenadas de los vectores de **B1** en la base **B2**.

■ Un ejemplo

Consideramos la base **B={(−1,0,1),(1,−1,0),(1,1,−1)}**

a) ¿Cuál es la matriz de cambio de base de **B** a la base canónica?

En este caso, como los vectores de **B** ya están dados en coordenadas de la base canónica, la matriz de cambio de base se forma poniendo por columnas los vectores de **B**:

```
In[5]:= CambioBaBC = Transpose[{{-1, 0, 1}, {1, -1, 0}, {1, 1, -1}}]

Out[5]= {{-1, 1, 1}, {0, -1, 1}, {1, 0, -1}}

In[6]:= MatrixForm[CambioBaBC]

Out[6]//MatrixForm=
```

$$\begin{pmatrix} -1 & 1 & 1 \\ 0 & -1 & 1 \\ 1 & 0 & -1 \end{pmatrix}$$

b) ¿Cuál es la matriz de cambio de base de la base canónica a **B**?

Este cambio de base es el contrario, luego su matriz de cambio de base es la inversa de la anterior:

```
In[7]:= CambioBCaB = Inverse[CambioBaBC]

Out[7]= {{1, 1, 2}, {1, 0, 1}, {1, 1, 1}}

In[8]:= MatrixForm[CambioBCaB]

Out[8]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 & 2 \\ 1 & 0 & 1 \\ 1 & 1 & 1 \end{pmatrix}$$

■ Otro ejemplo

Consideramos las bases **B1 = {(1,0,1), (0,1,1), (1,1,0)}** y **B2 = {(2,1,3), (0,1,0), (2,0,0)}**.

a) Determinar la matriz de cambio de base de **B1** a **B2**.

Podemos calcular esta matriz usando lo siguiente (**BC** representa la base canónica):

Matriz de cambio de **B1** a **B2** = (Matriz de cambio de **BC** a **B2**) . (Matriz de cambio de **B1** a **BC**)

```
In[9]:= B1 = {{1, 0, 1}, {0, 1, 1}, {1, 1, 0}};
        B2 = {{2, 1, 3}, {0, 1, 0}, {2, 0, 0}};

In[11]:= CambioB1aB2 = Inverse[Transpose[B2]].Transpose[B1]
```

Out[11]= $\left\{\left\{\frac{1}{3}, \frac{1}{3}, 0\right\}, \left\{-\frac{1}{3}, \frac{2}{3}, 1\right\}, \left\{\frac{1}{6}, -\frac{1}{3}, \frac{1}{2}\right\}\right\}$

```
In[12]:= MatrixForm[CambioB1aB2]
```

Out[12]//MatrixForm=

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{3} & 0 \\ -\frac{1}{3} & \frac{2}{3} & 1 \\ \frac{1}{6} & -\frac{1}{3} & \frac{1}{2} \end{pmatrix}$$

b) Determinar las coordenadas respecto de la base **B2** del vector **v** con coordenadas (1,2,3) en la base **B1**.

Estas coordenadas se hallan fácilmente usando la matriz de cambio de base que obtuvimos antes:

```
In[13]:= CambioB1aB2.{1, 2, 3}

Out[13]= {1, 4, 1}
```

c) Determinar la matriz de cambio de base de **B2** a **B1**.

Este cambio de base es el contrario al que hallamos antes, luego su matriz es la inversa:

```
In[14]:= CambioB2aB1 = Inverse[CambioB1aB2]
```

Out[14]= $\left\{\left\{2, -\frac{1}{2}, 1\right\}, \left\{1, \frac{1}{2}, -1\right\}, \left\{0, \frac{1}{2}, 1\right\}\right\}$

```
In[15]:= MatrixForm[CambioB2aB1]
```

Out[15]//MatrixForm=

$$\begin{pmatrix} 2 & -\frac{1}{2} & 1 \\ 1 & \frac{1}{2} & -1 \\ 0 & \frac{1}{2} & 1 \end{pmatrix}$$

■ Ejercicio

Calcula la matriz de cambio de base de **B2** a **B1** como en el apartado b) anterior, sin usar la matriz de cambio de **B1** a **B2**. (Comprueba que sale lo mismo que acabamos de obtener.)

■ Subespacios vectoriales

La orden **RowReduce** es interesante para el estudio de un subespacios vectorial del que conocemos un sistema de generadores, ya que nos da la dimensión del subespacio y una base del mismo.

■ Ejemplo

Consideramos el subespacio de $\boldsymbol{R}^4$ generado por los vectores (1, 2, −1,1), (0, 2, 4, 6), (3, 1, 1, 2), (2, 3, −1,2). Para calcular su dimensión y una base suya formamos la matriz que tiene por filas estos vectores y le aplicamos **RowReduce**:

```
In[16]:= U = {{1, 2, -1, 1}, {0, 2, 4, 6}, {3, 1, 1, 2}, {2, 3, -1, 2}}

Out[16]= {{1, 2, -1, 1}, {0, 2, 4, 6}, {3, 1, 1, 2}, {2, 3, -1, 2}}

In[17]:= MatrixForm[RowReduce[U]]
```

Out[17]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

Por tanto, la dimensión de **U** es 3 y una base suya es {(1,0,0,0), (0,1,0,1), (0,0,1,1)}.

■ Método de ortogonalización de Gram−Schmidt

Dada una base de $\boldsymbol{R}^n$ o de un subespacio suyo, es fácil calcular una base ortonormal del mismo espacio a partir de ella usando el proceso de ortogonalización de Gram−Schmidt que se estudia en teoría. *Mathematica* tiene órdenes que realizan este proceso automáticamente. Para poder usarlas es necesario ejecutar la siguiente órden, que carga en la memoria el paquete adecuado (cuidado con las comillas hacia atrás):

```
In[18]:= << LinearAlgebra`Orthogonalization`
```

■ Ejemplo

Consideramos la base de $\boldsymbol{R}^3$ dada por B={(1,−1,0),(0,1,−1),(1,0,1)}. Le aplicamos el método de Gram−Schmidt para obtener a partir de ella una base ortonormal:

```
In[19]:= GramSchmidt[{{1, -1, 0}, {0, 1, -1}, {1, 0, 1}}]
```

Out[19]= $\left\{\left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, 0\right\}, \left\{\frac{1}{\sqrt{6}}, \frac{1}{\sqrt{6}}, -\sqrt{\frac{2}{3}}\right\}, \left\{\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right\}\right\}$

■ Aproximación por mínimos cuadrados

En diversas situaciones prácticas es corriente disponer de un conjunto de datos experimentales y querer encontrar una función sencilla que se aproxime "lo mejor posible" a esos datos. Un posible criterio para decidir qué significa "lo mejor posible" viene dado por los ajustes por mínimos cuadrados. Veamos cómo usar *Mathematica* para encontrar la recta que mejor se aproxima a unos datos según este criterio (*ajuste lineal por mínimos cuadrados*) o la parábola que mejor se aproxima (*ajuste cuadrático por mínimos cuadrados*).

■ Ejemplo

Supongamos que tenemos los datos (1,4), (2,4), (3,5), (4,6). (Tal vez corresponden al número de visitantes de cierto museo en distintos días: el primer día, 4 visitantes; el segundo día, 4 visitantes...) Al conjunto de datos de que dispone uno a veces se le llama *nube de puntos*. Primero, introducimos los datos en dos vectores y los representamos:

```
In[20]:= x = {1, 2, 3, 4};
         y = {4, 4, 5, 6};
         n = Length[x];
         a = Min[x];
         b = Max[x];

In[25]:= datos = ListPlot[Table[{x[[i]], y[[i]]}, {i, n}],
           PlotStyle → {PointSize[0.03], RGBColor[0, 0, 1]}]
```




```
Out[25]= - Graphics -
```

■ Ajuste lineal

Para obtener la recta que mejor se aproxima a estos datos hacemos los siguiente (simplemente estamos siguiendo los cálculos que se deducen en teoría):

```
In[26]:= A1 = Table[x[[i]]^(j - 1), {i, n}, {j, 2}];
         MatrixForm[A1]
```

Out[27]//MatrixForm=

$$\begin{pmatrix} 1 & 1 \\ 1 & 2 \\ 1 & 3 \\ 1 & 4 \end{pmatrix}$$

```
In[28]:= a1 = Inverse[Transpose[A1].A1].(Transpose[A1].y)
```

Out[28]= $\left\{3, \frac{7}{10}\right\}$

```
In[29]:= f1 = a1.{1, z}    (* Esta es la recta que resulta finalmente *)
```

Out[29]= $3 + \frac{7\,z}{10}$

```
In[30]:= graf1 =
           Plot[f1, {z, a, b}, DisplayFunction → Identity, PlotStyle → {RGBColor[1, 0, 0]}];
```

```
In[31]:= Show[datos, graf1]
```




```
Out[31]= - Graphics -
```

■ Ajuste cuadrático

Obtenemos ahora la parábola que mejor se aproxima a los datos anteriores:

```
In[32]:= A2 = Table[x[[i]]^(j - 1), {i, n}, {j, 3}];
         MatrixForm[A2]
```

Out[33]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \end{pmatrix}$$

```
In[34]:= a2 = Inverse[Transpose[A2].A2].(Transpose[A2].y)
```

Out[34]= $\left\{\frac{17}{4}, -\frac{11}{20}, \frac{1}{4}\right\}$

```
In[35]:= f2 = a2.{1, z, z^2}   (* Esta es la parábola que buscamos *)
```

Out[35]= $\frac{17}{4} - \frac{11\,z}{20} + \frac{z^2}{4}$

```
In[36]:= graf2 =
          Plot[f2, {z, a, b}, DisplayFunction → Identity, PlotStyle → {RGBColor[0, 1, 0]}]

Out[36]= - Graphics -

In[37]:= Show[datos, graf2]
```




```
Out[37]= - Graphics -
```

```
In[38]:= Show[datos, graf1, graf2]
```




```
Out[38]= - Graphics -
```

■ Ejercicios

1−Decide si los siguientes conjuntos de vectores son linealmente independientes o no:

a) {(1,2,0), (1,2,1), (1,−2,0)}
b) {(3,4,0), (2,2,1), (1,2,2)}
c) {(1,2,1,1), (1,1,2,2), (1,2,4,1)}
d) {(1,2,1,2), (1,1,1,1), (2,2,5,6), (3,4,4,3)}

2−Los siguientes conjuntos, ¿son bases de $\boldsymbol{R}^4$?

a) {(1,2,3,5), (1,4,5,6), (5,6,7,8), (4,6,7,9)}
b) {(−3,−1,0,0)(0,1,1,−1),(2,2,5,6), (3,4,4,3)}

3−Consideramos en $\boldsymbol{R}^3$ los conjuntos de vectores siguientes:

$\boldsymbol{B}_1$= {(1,1,1), (1,1,2), (1,2,3)}, $\boldsymbol{B}_2$= {(2,1,2), (3,2,−5),(1,−1,1)}
a) Comprobar que son bases de R^3.
b) Calcular la matriz de cambio de base de $\boldsymbol{B}_1$ a $\boldsymbol{B}_2$.
c) Calcular la matriz de cambio de base de $\boldsymbol{B}_2$ a $\boldsymbol{B}_1$.
d) Calcular las coordenadas en la base $\boldsymbol{B}_1$ del vector cuyas coordenadas en $\boldsymbol{B}_2$ son (1,1,3).
e) Calcular una base ortonormal a partir de $\boldsymbol{B}_1$.
f) Calcular una base ortonormal a partir de $\boldsymbol{B}_2$.

4−Calcular la dimensión y una base del subespacio vectorial de $\boldsymbol{R}^4$ generado por los vectores {(1,1,1,1), (1,0,0,1), (0,1,1,2), (0,1,0,0)}.

5−Calcular la mejor aproximación lineal y cuadrática or mínimos cuadrados para el conjunto de datos siguiente: (1,2), (4,6), (9,3), (12,2), (22,11). Representar gráficamente los datos junto con las dos aproximaciones consideradas.